<commit_before>SUMMARY = "Updating Firmware in Linux"
LICENSE = "GPLv2"
LIC_FILES_CHKSUM = "file://${S}/COPYING;md5=b234ee4d69f5fce4486a80fdaf4a4263"

DEPENDS = "libgudev glib-2.0 polkit appstream-glib libgusb gpgme gcab-native intltool-native gettext-native"

SRC_URI = "git://github.com/hughsie/fwupd;method=https \
           file://meson-skip-test-directories-when-disabled.patch \
           "
SRCREV = "de3507d9c09f287570ff2de6c6c00b8c181a9f2f"
S = "${WORKDIR}/git"

# Introspection not working the way how meson does it (ends up calling the host ld).
EXTRA_OEMESON += "-Denable-introspection=false"

# Beware, some of the disabled features have dependencies for which
# there are no recipes.
PACKAGECONFIG ?= "${@ bb.utils.filter('DISTRO_FEATURES', 'systemd', d) } uefi libelf"
PACKAGECONFIG[colorhug] = "-Denable-colorhug=true,-Denable-colorhug=false,colorhug"
PACKAGECONFIG[consolekit] = "-Denable-consolekit=true,-Denable-consolekit=false,consolekit"
PACKAGECONFIG[doc] = "-Denable-doc=true,-Denable-doc=false,gtkdoc-native"
PACKAGECONFIG[dell] = "-Denable-dell=true,-Denable-dell=false,libsmbios_c"
PACKAGECONFIG[libelf] = "-Denable-libelf=true,-Denable-libelf=false,elfutils"
PACKAGECONFIG[man] = "-Denable-man=true,-Denable-man=false,docbook2man"
PACKAGECONFIG[systemd] = "-Denable-systemd=true,-Denable-systemd=false,systemd"
PACKAGECONFIG[thunderbolt] = "-Denable-thunderbolt=true,-Denable-thunderbolt=false,libtbtfwu"
PACKAGECONFIG[uefi-labels] = "-Denable-uefi-labels=true,-Denable-uefi-labels=false,cairo fontconfig"
PACKAGECONFIG[uefi] = "-Denable-uefi=true,-Denable-uefi=false,fwupdate,fwupdate"
# synaptics depends on dell
PACKAGECONFIG[synaptics] = "-Denable-synaptics=true,-Denable-synaptics=false"
PACKAGECONFIG[tests] = "-Denable-tests=true,-Denable-tests=false"

inherit check-available
inherit ${@ check_available_class(d, 'meson', ${HAVE_MESON}) }

do_install_append () {
    rm -rf ${D}/${datadir}/installed-tests
}

# Support groupcheck instead of polkit?
do_install_append_df-refkit-groupcheck () {
    # Instead of hard-coding actions, we take them from the polkit configuration.
    install -d ${D}${datadir}/groupcheck.d
    for action in `grep action.id ${D}${datadir}/polkit-1/*/*.policy | sed -e 's/.*"\(.*\)".*/\1/' | sort -u`; do
        # Just a sanity check that we really picked something that looks like an action.
        case $action in org.freedesktop.fwupd.*)
            # "adm" gets added as auxiliary group for root in base-passwd_%.bbappend.
            # We rely on that here to grant also root processes access to fwupd.
            echo $action='"adm"' >>${D}${datadir}/groupcheck.d/org.freedesktop.fwupd.policy;;
        esac<commit_msg>fwupd: gpg enabled, pkcs7 disabled<commit_after>SUMMARY = "Updating Firmware in Linux"
LICENSE = "GPLv2"
LIC_FILES_CHKSUM = "file://${S}/COPYING;md5=b234ee4d69f5fce4486a80fdaf4a4263"

DEPENDS = "libgudev glib-2.0 polkit appstream-glib libgusb gcab-native intltool-native gettext-native"

SRC_URI = "git://github.com/hughsie/fwupd;method=https \
           file://meson-skip-test-directories-when-disabled.patch \
           "
SRCREV = "de3507d9c09f287570ff2de6c6c00b8c181a9f2f"
S = "${WORKDIR}/git"

# Introspection not working the way how meson does it (ends up calling the host ld).
EXTRA_OEMESON += "-Denable-introspection=false"

# Beware, some of the disabled features have dependencies for which
# there are no recipes.
#
# gpg is needed to verify content downloaded from the LVFS. It can be
# disabled safely when firmware only gets delivered via the local filesystem.
PACKAGECONFIG ?= "${@ bb.utils.filter('DISTRO_FEATURES', 'systemd', d) } uefi libelf gpg"
PACKAGECONFIG[colorhug] = "-Denable-colorhug=true,-Denable-colorhug=false,colorhug"
PACKAGECONFIG[consolekit] = "-Denable-consolekit=true,-Denable-consolekit=false,consolekit"
PACKAGECONFIG[doc] = "-Denable-doc=true,-Denable-doc=false,gtkdoc-native"
PACKAGECONFIG[dell] = "-Denable-dell=true,-Denable-dell=false,libsmbios_c"
PACKAGECONFIG[libelf] = "-Denable-libelf=true,-Denable-libelf=false,elfutils"
PACKAGECONFIG[man] = "-Denable-man=true,-Denable-man=false,docbook2man"
PACKAGECONFIG[systemd] = "-Denable-systemd=true,-Denable-systemd=false,systemd"
PACKAGECONFIG[thunderbolt] = "-Denable-thunderbolt=true,-Denable-thunderbolt=false,libtbtfwu"
PACKAGECONFIG[uefi-labels] = "-Denable-uefi-labels=true,-Denable-uefi-labels=false,cairo fontconfig"
PACKAGECONFIG[uefi] = "-Denable-uefi=true,-Denable-uefi=false,fwupdate,fwupdate"
# synaptics depends on dell
PACKAGECONFIG[synaptics] = "-Denable-synaptics=true,-Denable-synaptics=false"
PACKAGECONFIG[tests] = "-Denable-tests=true,-Denable-tests=false"
PACKAGECONFIG[gpg] = "-Denable-gpg=true,-Denable-gpg=false,gpgme"
PACKAGECONFIG[pkcs7] = "-Denable-pkcs7=true,-Denable-pkcs7=false,gnutls"

inherit check-available
inherit ${@ check_available_class(d, 'meson', ${HAVE_MESON}) }

do_install_append () {
    rm -rf ${D}/${datadir}/installed-tests
}

# Support groupcheck instead of polkit?
do_install_append_df-refkit-groupcheck () {
    # Instead of hard-coding actions, we take them from the polkit configuration.
    install -d ${D}${datadir}/groupcheck.d
    for action in `grep action.id ${D}${datadir}/polkit-1/*/*.policy | sed -e 's/.*"\(.*\)".*/\1/' | sort -u`; do
        # Just a sanity check that we really picked something that looks like an action.
        case $action in org.freedesktop.fwupd.*)
            # "adm" gets added as auxiliary group for root in base-passwd_%.bbappend.
            # We rely on that here to grant also root processes access to fwupd.
            echo $action='"adm"' >>${D}${datadir}/groupcheck.d/org.freedesktop.fwupd.policy;;
        esac
{'fraction_non_alphanumeric': 0.13966971187631763, 'fraction_numerical': 0.017919887561489812, 'mean_word_length': 4.967505241090147, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 7, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 1, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1082680', 'n_tokens_mistral': 2123, 'n_tokens_neox': 2098, 'n_words': 389}